
Abstract. We present a case study on formalization of a textbook the-
orem in a form that is as close to the original textbook presentation as
possible. Euler's partition theorem, listed as #45 at Freek Wiedijk's list
of �Top 100 mathematical theorems�, is taken as the subject of the study.
As a result new formal concepts including informal �exary (i.e. �exible
arity) addition are created and existing ones are extended to go around
existing limitations of the Mizar system, without modi�cation of its core.
Such developments bring more �exibility of informal language reasoning
into the Mizar system and make it useful for wider audience.

1 Introduction

Famous mathematical theorems rarely occur with only one proof in informal
mathematical practice. In the mathematical literature we can often �nd several
formulations or even conceptually di�erent proofs of the same theorem. How-
ever, the reader can easily compare proofs that have the same main idea. Such
situations are not so popular in repositories of formal mathematical knowledge.
Usually, one version of a theorem with one proof only is stored there. Addition-
ally, comparing proofs created in di�erent formal proof systems is not so trivial.
Even, if we consider two declarative environments or two procedural ones, this
problem does not seem much easier.

It comes as no surprise that the main idea of the formal proof is often dif-
ferent from all known informal proof variants of the theorem, even if the author
tried to create a formal equivalent of a particular informal development. The
experience of big proof formalization developments shows that proof script au-
thors can often, given the set of de�nitions and theorems collected in the Mizar
Mathematical Library (MML) [3], obtain a new, so far unknown, and sometimes
simpler, proof of a particular statement [8, 9]. Therefore, many authors compare
informal proof variants to check the possible use of collected resources before
starting their formalization e�ort.

? The paper has been �nanced by the resources of the Polish National Science Center

2 K. P¡k

To illustrate such situations we can consider the development of Brouwer's
�xed-point theorem [5]. This theorem has been used in the formalization of
Jordan curve theorem in the Mizar [7]. But for this purpose, the 2-dimensional
case was enough. Therefore, this statement has been provided by A. Korniªowicz,
only for this case using basic arguments concerning the fundamental groups
of the respective spaces [13]. Note that this approach for higher-dimensional
cases requires incomparably more di�cult facts about these groups. The same
theorem was proved in a combinatorial way in the HOL Light by J. Harrison
for n-dimensional case, based on Sperner's lemma [10]. The original approach
to prove this lemma is based on intuitively clear facts about the standard n-
dimensional simplex and its arbitrarily small subdivision. However, these facts
are not so easy if we consider them formally. Therefore, he chose an alternative
justi�cation of this lemma, wherein the simplex structure is replaced by the cubic
one. Note that simplex structure is explored in one of the approaches to prove
Brouwer's invariance of the domain theorem that was selected to formalize in
Mizar by K. P¡k [18]. Therefore, having a large collection of facts about simplices,
Brouwer's �xed-point theorem has been redeveloped to the general case based
on Sperner's lemma in the original approach.

In this paper, we present the results of an experiment where we formalize
Euler's partition theorem in the original approach [1, 6, 22]. Obviously, we can
obtain a very slick proof using de�nitions and theorems collected in the MML
that looks more or less similar to the original proof. However, the point of this
exercise was not to obtain �a formalization�, but to see how a natural language
proof can be expressed in the Mizar format. Therefore our aim was to recreate
the main idea and steps of reasoning as closely as possible, sometimes work
around the system's limitations, however without a modi�cation of its core, to
obtain the result that looks almost the same as the informal one. Furthermore,
as a measure of �closeness�, we consider also the sketch of the proof that is
generated automatically from the Mizar proof scripts and is published in the
journal Formalized Mathematics.

Structure of the paper In Section 2 we discuss several conceptually di�erent
proofs of Euler's partition theorem. We focus our attention on three approaches:
the original one that was presented by L. Euler [6], the Euler's bijective proof
that was presented by G.E. Andrews [1], and the approach basing on Sylvester's
bijection created by J.J. Sylvester, and choose one. In Section 3 we analyze
informal mathematical constructions that are used in the selected approach, and
we propose an adaptation method of this construction to the formal language
in a way that the obtained visual e�ect is as closely as possible to the informal
one. In Section 4 we present a formalization of the proof in the selected approach
written in the Mizar system. Finally, in Section 5 we conclude the paper and
we discuss future work. Note additionally that each fragment of the Mizar proof
scripts contained in this paper comes from Mizar theory �les FLEXARY1.miz,
EULRPART.miz available in the Mizar distribution.

2 Informal proofs of Euler's partition theorem

Generally, a partition of a natural number n is a way of writing n as a sum
of positive integers where the arrangement of the addends does not need to be
determined. Denote by On the set of partitions of n into odd parts and similarly
denote by Dn the set of partitions of n into distinct parts. Then Euler's partition
theorem states that the cardinality of On is equal to the cardinality of Dn for all
natural n. Euler presented a very slick proof in 1748 [6] by generating functions
that can be sketched as follows:

1 +

∞∑
n=1

|Dn|xn = (1 + x) · (1 + x2) · (1 + x3) · (1 + x4) · (1 + x5) · . . .

=
1− x2

1− x
· 1− x

4

1− x2
· 1− x

6

1− x3
· 1− x

8

1− x4
· 1− x

10

1− x5
· . . .

=
1

1− x
· 1

1− x3
· 1

1− x5
· . . .

= (1 + x+ x2 + x3 + . . .) · (1 + x3 + (x3)2 + (x3)3 + . . .)
· (1 + x5 + (x5)2 + (x5)3 + . . .) · . . .

= 1 +

∞∑
n=1

|On|xn

(1)

However, in such an analytical proof, information that describes the relationship
between relevant partitions of n is implicit and hard to grasp. Obviously, without
this information the proof is complete, but there are many people who prefer
to compare the cardinality of sets based on an explicit mapping that associates
their elements. Such a bijective proof has also been described by Euler. It has
been given by G.E. Andrews [1, pp. 149-150], and also by H.S. Wilf [22, p. 10]
in the following form:

Euler's bijective proof: A partition into distinct parts can be written as

n = d1 + d2 + . . .+ dk. (3)

Each integer di can be uniquely expressed as a power of 2 times an odd number. Thus,
n = 2a1O1 + 2a2O2 + 2a3O3 + . . . + 2akOk where each Oi is an odd number. If we now
group together the odd numbers we get an expression like:

n = (2α1 + 2α2 + . . .) · 1 + (2β1 + 2β2 + . . .) · 3 + (2γ1 + 2γ2 + . . .) · 5 + . . .
= µ1 · 1 + µ3 · 3 + µ5 · 5 + . . .

In each series (2α1 + 2α2 + . . .), the αi's are distinct (why?). Thus the sum is the binary
expansion of some µj . We now see the partition of n into odd parts that corresponds, under
this bijection, to the given partition (3) into distinct parts. It is the partition that contains
µ1 1's, µ3 3's, etc. �

Fig. 1. A bijective proof of Euler's partition theorem that is used in [22].

In this constructive proof only several simple facts are used (implicite): the
existence and the uniqueness of conversion between a natural number and its

binary equivalent; a positive number decomposition as the product of a power
of two and an odd number.

It is also possible to prove this result by another partition transformation,
where correspondence is visually apparent by some modi�cation of the Ferrers
diagram, called a bent graph. The construction of this graph is based on the
observation that we can represent every odd number 2i+1 as one central point,
and a column and a row that are built with i points. Then we may arrange the
con�gurations of points in such a way that the central points are inserted into
a diagonal line. In this way we obtain what we call the bent graph (see Fig. 2).
Additionally, the symmetrical parts of this graph, located above and below the
diagonal line, are called regular graphs. Sylvester's bijection that is de�ned on
bent graphs has been presented in [20, pp. 287�288] and is formulated as follows:
Each of these graphs will be bounded by lines inclined to each other at an angle
one-half of that contained between the original bounding lines, and each may be
regarded as made up of bends �tting into one another. The original proof also
contains a sketch of a justi�cation that the resulting partition consists of pairwise
di�erent numbers (see to the example presented in Fig. 2 as an illustration).
Obviously, the justi�cation presented in this form shows in a simple way the

11 11 9 5 5 5 2 8 10 11

9

6

Fig. 2. An example that illustrates Sylvester's bijection, which maps a partition of 46
into odd numbers (5, 5, 5, 9, 11, 11) to a partition of 46 into pairwise di�erent numbers
(2, 6, 8, 9, 10, 11), used in [20].

proof idea, but only the idea. Therefore, a formalization of such a reasoning
could not re�ect the original proof at a very high level of similarity.

Analysis of well-known justi�cations of Euler's partition theorem for the pur-
poses of formal transcription shows that re�ecting the original proof is possible
in two mentioned approaches. However, the second justi�cation, presented in
Fig. 1, contains a more interesting informal construction, namely �exary (i.e.
�exible arity, see [12]) addition with visible lower and upper bounds of summa-
tion, but also with only a lower bound available. Additionally, the formalization

of this theorem in the HOL system imitates also the main idea of the second
approach1.

3 Formal introduction of informal notations

To re�ect the informal reasoning in the Mizar language, we discuss in this section
de�nitions and notations used in the reasoning presented in Fig. 1. We divide
this discusion into three subsections. In Subsection 3.1 we choose concepts of a
partition de�nition. Next, in Subsection 3.2 we de�ne operators that re�ect in
the Mizar system the informal �exary plus in both cases, �nite and in�nite. At
the end, in Subsection 3.3 we de�ne a special kind of a matrix generalization to
realize the rearrangement of the values of a �nite sequence into a sequence of
sequences that is used in the considered reasoning.

3.1 The formal de�nition of a partition

The reasoning presented in Fig. 1 uses a partition of a natural number n (see (3))
de�ned as a �nite sequence of positive integers that sum up to n. Note also that
the order of the addends is not indicated. However, to count partitions of a
number we need to opt for some type of arrangement, non increasing or non
decreasing. The alternative solution is to represent a partition as a sequence
of addends frequency, i.e. a sequence (a1, a2, . . .) that represent the partition
{1, 1, . . . , 1︸ ︷︷ ︸

a1

, 2, 2, . . . , 2︸ ︷︷ ︸
a2

, . . .}. Obviously, we �nd this approach in the formulation

�the partition that contains µ1 1's, µ3 3's� where the non-decreasing arrangement
is suggested. Therefore, to simplify, we use only one method of arrangement in
the de�nition of partition, which is formulated as a non-decreasing �nite sequence
of non-zero natural numbers that sum up to n, thus obtained by writing (in
Mizar)

definition

let n be Nat;
mode a_partition of n -> non-zero non-decreasing natural-valued

FinSequence means

Sum it = n;
end;

(2)

Obviously, such a de�nition is adapted to represent simple modi�cations of parti-
tions in an intuitive way, but requires a special attention in the formal approach.
Note that we can neither simply include elements to a partition, nor modify its
existing elements, without violating the arrangement. Such problems do not oc-
cur if we consider the frequency representation of partitions. In this case the
realization of the mentioned operations on a partition is reduced to a simple
modi�cation that increases or reduces by one the value of one element in the
frequency sequence. This approach has been used in the formalization of Euler's
theorem in HOL, where the partition is de�ned as follows:

1 For more details see https://code.google.com/p/hol-light/source/browse/

trunk/100/euler.ml?r=2.

let partitions = store_name "partitions" new_definition

`p partitions n <=> (!i. ~(p i = 0) ==> 1 <= i /\ i <= n) /\

nsum(1..n) (\i. p(i) * i) = n`;;

3.2 The �exary plus

Analyzing the �rst sentence of the proof, we �nd the equation (3) where we
face two formalization problems. Obviously, the informal mathematical operators
are present here, i.e. the �exary plus. This formula is usually formalized in the

equivalent form n =

k∑
i=1

di. It can be written in the Mizar language as n = sum d,

if we additionally assume that k is the length of d, or it can be written as
n = sum (d|k), where k is an arbitrary natural number, d is a �nite sequence of
natural numbers, and d|k is the restriction of d to the set {1, 2, . . . ,k}. However,
in our experiment we want to obtain a similar term represented as:

(d,2)+...+(d,k). (4)

Note also that the informal expression d2+. . .+dk contains a hidden information
that the �nite sequence d has a second, a third, and up to a k-th element in
the domain. The method of hiding this information in a term as (4) or getting
around this problem is the second formalization problem. In the Mizar language
we can resolve this, e.g. by creating a de�nition by cases or summarizing only
these values that correspond to arguments in the domain of a �nite sequence. In
our experiment, we use a solution that gives the greater �exibility. The solution
is based on the concept developed in the MML: a permissive de�nition of the
function value, where it is assumed that the empty set is the value of a function
outside its �eld, and also on the Peano number approach, where the empty set
equals 0. However, this solution can be applied to summation of such D-valued
sequence if D contains 0. We de�ne the �exary plus as follows:

definition

let k,n;
let f,g be complex-valued Function;
func (f,k) +...+ (g,n) -> complex number means

h.(0+1) = f.(0+k) & ... & h.(n-’k+1) = f.(n-’k+k)
implies

it = Sum (h| (n-’k+1)) if f = g & k <= n
otherwise it = 0;

end;

(5)

where h is a complex valued �nite sequence, the operation -’ is the limited
substraction of natural numbers, i.e. a-’b is equal to max{a− b, 0}, & ... & is
the �exary logical conjunction (for more details see [11]). Note that the Mizar
system's limitations prohibit the repetition of a locus in an operator expression
when it is de�ned. Therefore, we cannot eliminate the function g, even if we
want to consider only the case f = g.

The value of the de�ned above �exary plus is a complex number, but in the
reasoning presented in Fig. 1 only the natural valued �nite sequences are used,

for which this value should be a natural number. To obtain such information
about the value, the su�cient solution is to have the following registration in
the environment of the Mizar article:

registration

let n,k;
let f be natural-valued FinSequence;
cluster (f,k) +...+ (f,n) -> natural;

end;

(6)

Based on the �exary plus de�nition, we can formulate and prove the �rst equal-
ity presented in Fig. 1 as n = d.1 + (d,2)+...+(d,len d) for an arbitrary
partition d of n, where len d is the length of d. Additionally, if we consider
�nite sequences a, O that represent the unique decomposition of d as a power of
2 times an odd number, we can formalize also the second equality as:

n = 2|^(a.1) * O.1 + 2|^(a.2) * O.2 +
(O (#) 2|^a,3) +...+ (O (#) 2|^a,len d),

(7)

where (#) represents the product of functions and a, O have already been intro-
duced to the reasoning in the preceding step which reads as follows:

consider O be odd-valued FinSequence,
a be natural-valued FinSequence such that

A1: len O = len d = len a & d = O (#) 2|^a and

A2: d.1 = O.1*(2|^a.1) &...& d.len d = O.len d*(2|^a.len d);

(8)

Observe that the formula labeled by A2 is an equivalent formulation of the
statement d = O (#) 2|^a. This statement has been added only for improving
the readability of dependencies occurring between d, O, and a. Note also that
we could prove the equality (7) without the restriction on the length of d (that
is equal to the length of O and a), e.g. for len O equal 0 we obtain simply that
O.1 = O.2 = 0, since 1, 2 do not belong to the domain of O, but also n = 0,
since 0-length sequence can be only the partition of 0.

Analyzing the next part of the reasoning presented in Fig. 1 we can observe
that the �exary plus is used also in an unbounded form, without the upper
bound of summation. Generally, this operation is used in the informal mathe-
matical practice to speak conveniently about the sum of the terms of a sequence,
where basing on several �rst terms we can precisely predict the others elements
by analogy. Additionally, according to a popular informal convention, the infor-
mation about the convergence of a sequence is often assumed a priori. However,
this issue does not concern the reasoning presented in Fig. 1, where the un-
bounded �exary plus is used only in the context of �nite sequences. For such
kind of sequence, we can de�ne this operator as the �exary plus with an upper
bound, where the upper bound is greater than or equal to the maximum of the
domain of the sequence. It is obtained by writing:

definition

let n;
let f be complex-valued Function;
assume dom f /\ NAT is finite;
func (f,n)+... -> complex number means

for k st for i st i in dom f holds i <= k holds

it = (f,n) +...+ (f,k);
end;

(9)

Note that in the considered reasoning, only �nite sequences are used, where
the intersection of the domain and N is �nite. Therefore, the introduction of
such a de�nition to our experiment seems to be redundant. However, without
this assumption, we cannot use this operator in the Mizar system if we need to
substitute a term that is a priori a �nite sequence on f, but we have not proved
this statement yet. Obviously, such a possibility is very useful, if we want to
formulate similarly a formal equivalents of an informal term. Moreover, we can
reinforce the Mizar checker in such a way that the equality

(f,n)+... = (f,n) +...+ (f,len f); (10)

is automatically generated and added to every justi�cation of a step, where the
expression +... is used and a term substituted for f is a �nite sequence. For
this purpose we create the following rede�nition:

definition

let n;
let f be complex-valued FinSequence;
redefine func (f,n)+... -> complex number equals

(f,n) +...+ (f,len f);
end;

(11)

We are aware that a de�nition that can generate automatically the summed
sequence based only on the terms on endpoints (in a �nite case) or two consec-
utive terms (in an in�nite case) is a more interesting solution. However, such a
solution requires a modi�cation of the core Mizar system, as it has been done in
the case of �exary logical operators for generalized conjunction and alternative
(for more details see [11]). Such a solution goes beyond the point of this study.

3.3 Regrouping the values of sequence

In the considered reasoning, we come across another interesting informal proce-
dure that consists of grouping the odd number. Obviously, partitioning of a set
into non-empty subsets, according to some properties of its members is noth-
ing new. However, to improve the readability of the de�ned partition, the proof
authors add to the reasoning some exempli�cations or even write elements of
several members in such a way that the reader can easily �nd out other mem-
bers by analogy. Note that such kind of exempli�cation is important for humans,
but generally is unnecessary for the Mizar checker, except from the case where
the existence of some kind of partition is proved.

Therefore, specially for our experiment, we de�ne a speci�c kind of a �nite
sequence of �nite sequences (a matrix generalization) over an odd valued �nite
sequence O, denoted in the Mizar language by odd_organization of O. This
map is de�ned in such a way that the �rst sequence contains all arguments of O
for which the value equal 1, the second sequence contains all arguments of O for
which the value equal 3, etc., where the number of �nite sequences is su�cient
to cover the domain of O. Obviously, odd_organization of O is not uniquely
determined by O, but we can use the global choice [14] (for more details see [7]).

Note that we can de�ne odd_organization �directly�, i.e. without sub-
types and attributes. However, if another user of the MML will need to regroup
the values in a di�erent way, then probably he would have to provide some anal-
ogous properties. To avoid such situations in the MML, new types are de�ned
as a restricted version of existing, more general types, if only the last ones exist.
Therefore, we de�ne the odd_organization in the following way. First we
note that individual �nite sequences in odd_organization have to be injec-
tive and determined, disjoint sets of values. Hence we introduce the following
attribute:

definition

let F be Function-yielding Function;
attr F is double-one-to-one means

for x1,x2,y1,y2 be object st

x1 in dom F & y1 in dom (F.x1) &

x2 in dom F & y2 in dom (F.x2) & F_(x1,y1)=F_(x2,y2)
holds x1 = x2 & y1 = y2;

end;

(12)

and a mode that reorganizes a �nite set D into a �nite sequence of �nite se-
quences:

definition

let D be finite set;

mode DoubleReorganization of D -> double-one-to-one FinSequence of D*
means Values it = D;

end;

(13)

Then we de�ne a type where we have that in every individual �nite sequence, ele-
ments are mapped to the same value, and such values in di�erent �nite sequences
are di�erent:

definition

let f be finite Function;
mode valued_reorganization of f -> DoubleReorganization of dom f means

(for n ex x st

x = f.it_(n,1) & ... & x = f.it_(n,len (it.n))) &

for n1,n2,i1,i2 be Nat st

i1 in dom (it.n1) & i2 in dom (it.n2) &

f.it_(n1,i1) = f.it_(n2,i2)
holds n1 = n2;

end;

(14)

and �nally we de�ne odd_organization as follows:
definition

let f be odd-valued FinSequence;
mode odd_organization of f -> valued_reorganization of f means

2*n-1 = f.it_(n,1) & ... & 2*n-1 = f.it_(n,len (it.n));
end;

(15)

Based on this approach, we can prove in a more general form the properties
of odd_organization that are needed to justify steps in the considered the-
orem. Obviously this approach is much more di�cult, but is consistent with
the popularized direction of the development of the MML. According to this
direction, the legible formulation and proving of a theorem is an important and
challenging aim, when proof scripts are created for further development of the
MML. However, no less important in this direction is extraction of de�nitions,
creation of auxiliary theorems and notations in such a way that MML users will
be able to adapt this knowledge for their own purposes.

4 The theorem formalization

In the reasoning presented in Fig. 1 the �rst and also the biggest part is the
description of the transformation that maps a partition of a number into odd
parts to a partition of the number into distinct parts. To adapt this fragment in
a Mizar proof script we de�ne this transformation as follows:

definition

let n be Nat;
let p be one-to-one a_partition of n;
func Euler_transformation p -> odd-valued a_partition of n means

(16)

where the value denoted by it can be determined by the condition:
for O be odd-valued FinSequence,a be natural-valued FinSequence,

sort be odd_organization of O st

len O = len p = len a & p = O (#) 2|^a
for j holds card Coim(it,j*2-1) = ((2|^a)*.sort.j,1)+...

(17)

However, we decided on a more descriptive de�nition, where several formulas
occur that describe some �exempli�cations�, only to improve the legibility of
obtained condition. Note that the equivalence of such extended de�nition, pre-
sented below, with above ones is provided [15, (12)].

for j be Nat,O1 be odd-valued FinSequence,a1 be natural-valued FinSequence st

len O1 = len d = len a1 & d = O1 (#) 2|^a1
for sort1 be DoubleReorganization of dom d st

(1 = O1.sort1_(1,1) & ... & 1 = O1.sort1_(1,len (sort1.1))) &

(3 = O1.sort1_(2,1) & ... & 3 = O1.sort1_(2,len (sort1.2))) &

(3 = O1.sort1_(3,1) & ... & 5 = O1.sort1_(3,len (sort1.3))) &

for i holds

2*i-1 = O1.sort1_(i,1) & ... & 2*i-1 = O1.sort1_(i,len (sort1.i))
holds

card Coim(it,1) = (2|^a1).sort1_(1,1)+((2|^a1)*.sort1.1,2)+... &

card Coim(it,3) = (2|^a1).sort1_(2,1)+((2|^a1)*.sort1.2,2)+... &

card Coim(it,5) = (2|^a1).sort1_(3,1)+((2|^a1)*.sort1.3,2)+... &

card Coim(it,j*2-1) = (2|^a1).sort1_(j,1)+((2|^a1)*.sort1.j,2)+...

(18)

Obviously to prove the correctness of this de�nition in the Mizar system,
we have to justify that such a value exists and is unique. We selected the proof
of the �rst conditions [15, (11)] to formally represent the informal description
of Euler's transformation. For this aim, we constructed a reasoning, wherein
all steps that are located on the �rst level of nesting (for more details see [7])
correspond to the selected fragments of the informal proof. Additionally, as a
measure of correspondence we can analyze the generated automatically sketch
of this reasoning presented in Fig. 3. Note that the full proof contains about 300
lines (for the full description see [15]), therefore we hide all nested reasonings
and every list of statements that is used in a justi�cation.

(11) Let us consider a one-to-one partition d of n. Then there exists an odd-valued par-
tition e of n such that for every natural number j for every odd-valued �nite sequence
O1 for every natural-valued �nite sequence a1 such that lenO1 = len d = len a1 and
d = O1 ·2a1 for every double reorganization τ of dom d such that 1 = O1(τ1,1) and ... and
1 = O1(τ1,len(τ(1))) and 3 = O1(τ2,1) and ... and 3 = O1(τ2,len(τ(2))) and 5 = O1(τ3,1)
and ... and 5 = O1(τ3,len(τ(3))) and for every i, 2 · i− 1 = O1(τi,1) and ... and 2 · i− 1 =

O1(τi,len(τ(i))) holds Coim(e, 1) = 2a1 (τ1,1) + ((2a1 � τ)(1), 2)+ . . . and Coim(e, 3) =

2a1 (τ2,1) + ((2a1 � τ)(2), 2)+ . . . and Coim(e, 5) = 2a1 (τ3,1) + ((2a1 � τ)(3), 2)+ . . . and

Coim(e, j · 2− 1) = 2a1 (τj,1) + ((2a1 � τ)(j), 2)+ . . ..
Proof: n = d(1) + ((d, 2)+ . . .+(d, len d)) by [16, (22)]. Consider O being an odd-valued
�nite sequence, a being a natural-valued �nite sequence such that lenO = len d = len a
and d = O · 2a and d(1) = O(1) · 2a(1) and ... and d(len d) = O(len d) · 2a(len d). n =

2a(1)·O(1)+2a(2)·O(2)+((O·2a, 3)+ . . .+(O·2a, len d)) by [16, (20)], [21, (25)]. Reconsider
σ = the odd organization of O as a double reorganization of dom2a. Consider µ being a
(2 · lenσ)-element �nite sequence of elements of N such that for every j, µ(2 · j) = 0
and µ(2 · j − 1) = 2a(σj,1) + ((2a � σ)(j), 2)+ . . .. Set α = a · σ(1). Set β = a · σ(2).
Set γ = a · σ(3). n = (2α(1) + (2α, 2)+ . . .) · 1 + (2β(1) + (2β , 2)+ . . .) · 3 + (2γ(1) +
(2γ , 2)+ . . .) · 5 + ((iddomµ · µ), 7)+ . . . by [21, (29)], [16, (41)], [21, (25)], [4, (12)]. n =
µ(1)·1+µ(3)·3+µ(5)·5+((iddomµ ·µ), 7)+ . . . by [16, (42), (41), (25)]. Consider e being an

odd-valued �nite sequence such that e is non-decreasing and for every i, Coim(e, i) = µ(i).

n = Coim(e, 1) ·1+ Coim(e, 3) ·3+ Coim(e, 5) ·5+((iddomµ ·µ), 7)+ . . .. n =
∑
C by [16,

(20)], (9). For every j such that 1 ≤ j ≤ len d holds O(j) = O1(j) and a(j) = a1(j) by [21,

(25)], [19, (9)], [2, (4)]. For every j, Coim(e, j · 2− 1) = 2a1 (τj,1) + ((2a1 � τ)(j), 2)+ . . .
by [16, (42)], [21, (29)], [4, (72)], [16, (22)]. �

Fig. 3. The sketch generated automatically of proof [15, (11)] that justi�es the existence
of Euler_transformation. The content of the sketch has not been changed with
the exception of the order of bibliography items.

At the beginning, we introduce a partition d of n into distinct parts and
represent that its elements sum up to n:

let d be one-to-one a_partition of n;
n = d.1 + (d,2)+...+(d,len d) proof...

(19)

Then we introduce two �nite sequences that describe a decomposition of every
element of partition p as the product of a power of two and an odd number. We

formalize also the informal connection between these �nite sequences and the
number n that occur in the considered reasoning.

consider O be odd-valued FinSequence,a be natural-valued FinSequence such that

len O = len d = len a & d = O (#) 2|^a and

d.1 = O.1 * (2|^a.1) & ... & d.len d = O.len d * (2|^a.len d) by...
n = 2|^(a.1) * O.1 + 2|^(a.2) * O.2 + (O (#) 2|^a,3) +...+(O (#) 2|^a,len d)

proof...

(20)

As it has been mentioned in Section 3.3 to select a method that groups together
values of O we use the global choice: the odd_organization of O. Note that
the same method is used to reorganize the values of 2|^a. It can be done since
lengths of 2|^a and O are equal. However, we have to include this information
in the type of this reorganization. In the Mizar system, such modi�cation of the
type can be realized as follows:

len (2|^a)=len O by...
then reconsider sort = the odd_organization of O as

DoubleReorganization of dom (2|^a) by...
(21)

To formalize the unlabelled equality presented in Fig. 1 we have to introduce a
�nite sequence and three sets:

consider mu be (2*len sort)-element FinSequence of NAT such that

for j holds mu.(2*j) = 0 &

mu.(2*j-1) = (2|^a). sort_(j,1) + ((2|^a)*.sort.j,2)+... by...
set alpha = a*(sort.1), beta = a*(sort.2), gamma = a*(sort.3);

(22)

Then this equality can be formally formulated as follows:

n = ((2|^alpha).1+ (2|^alpha,2)+...) * 1 + ((2|^beta).1+ (2|^beta,2)+...) * 3 +
((2|^gamma).1+ (2|^gamma,2)+...) * 5 + ((id dom mu)(#)mu,7)+... proof...

n = mu.1 * 1 + mu.3 * 3 + mu.5 * 5 + ((id dom mu)(#)mu,7) +... proof...

(23)

Finally, to �nish the informal contraction of value of p in Euler's transformation,
we use the following 5 steps:

consider e be odd-valued FinSequence such that
e is non-decreasing & for i holds card Coim(e,i) = mu.i by...

n = card Coim(e,1) * 1 + card Coim(e,3) * 3 +
card Coim(e,5) * 5 + ((id dom mu)(#)mu,7) +... proof...

n = Sum e proof...
then reconsider e as odd-valued a_partition of n by...
take e;

(24)

However, these steps are not su�cient to �nish a formal proof of the existence.
For this purpose we need also to provide that the choice of e does not depend
on a, O and sort. In the considered reasoning a part of this information is
mentioned as �di can be uniquely expressed as a power of 2 times an odd number�.
We recreate this information proving that for every pair of �nite sequences a1,
O1 that satis�es d = O1 (#) 2|^a1 holds for j st 1 <= j & j <= len d holds

O.j = O1.j & a.j = a1.j. Whereas the in�uence of sort selection for obtained
value e is omitted. Therefore, to �nish the proof, we provide this in�uence as an
external auxiliary theorem [15, (5)] and only refering to this information from
nesting reasoning that justi�es the following proof step:

for j holds

card Coim(e,j*2-1) = (2|^a1).sort1_(j,1) + ((2|^a1)*.sort1.j,2)+...
(25)

where sort1 is an arbitrary odd_organization of O1.
Obviously, in the situation described above, where only several steps at the

�rst level of nesting do not have their informal counterparts in the text book
proof, it has a negative consequence, i.e. the size of the full proof. There are
three main reasons for that. Firstly, note that to obtain such a proof we encap-
sulate less important fragments of reasoning at the deeper levels of nesting as
subreasonings. Additionally, if steps in two di�erent nesting subreasonings refer
to the same auxiliary fact, then to avoid duplication, generally, we try to locate
this fact in a common top level of nesting in such a way that this fact is avail-
able from these subreasonings (for more details see [17]). However, this solution
is inconsistent with our goal, where we try to remove auxiliary facts form the
�rst level of nesting. The second reason is the consequence of faithfully repro-
ducing the informal term without any restrictions that are a priori assumed in
the informal context. Note that to resolve the problem of a priori assumptions
we have to provide several facts for di�erent cases, that are completely redun-
dant if we resign from the mirroring. Let us focus on the term 2α1 + 2α2 + . . .
that occurs in Fig. 1 and can be formalized simply as sum (2|^alpha). But
according to our purpose, we would like to obtain the term 2|^(alpha.1) +
(2|^alpha,2)+... that unfortunately is equal to 1 (= 2|^0), if alpha is the
empty �nite sequence, where at the same time sum (2|^alpha)=0. Therefore,
to resolve this problem we use (2|^alpha).1 instead of 2|^(alpha.1) in
(23). The third reason is related to the previous one. Note that premises where
we extract �rst few terms of summations are easily readable for a human, but
di�cult to use as premises. Often to use such premises we have to insert back
these extracted terms and consider again the above-mentioned redundant cases.

This shows that the adaptation of the main idea from an informal proof to
a formal one is not so trivial if we want to recreate it in a very precise way.

Since we de�ned Euler_transformation, we can prove that this trans-
formation is a bijection. Note that the textbook proof contains only very sketchy
justi�cation of this property, that is formulated as follows: In each series (2α1 +
2α2 + . . .), the αi's are distinct (why?). We provide this fact in the following
form [15, (13)]:

for O be odd-valued FinSequence, a be natural-valued FinSequence,
s be odd_organization of O st len O = len a & O (#) 2|^a is one-to-one

holds a*.s.i is one-to-one
(26)

We remind that s.i is the �nite sequence of all elements of the domain of O, for
which the value under O is equals to 2*i-1. Then a*.s.i is the image of the

elements of s.i under a. Based on this fact and the uniqueness of the binary
number representation we provide in [15, (14)] that Euler_transformation
is an injection:

for p1,p2 be one-to-one a_partition of n st

Euler_transformation p1 = Euler_transformation p2
holds p1 = p2

(27)

Obviously, we provide also in [15, (15)] that Euler_transformation is a
surjection, based on the existences of the binary number representation.

for e be odd-valued a_partition of n
ex p be one-to-one a_partition of n st

e = Euler_transformation p
(28)

Based on the two above-described statements we can �easily� prove that the
set of all natural valued �nite sequences that are partitions of n into odd parts has
the same size as the set of all natural-valued �nite sequences that are partitions
of n into distinct parts. This statement can be represented as follows:

card {p where p is Element of NAT*:p is odd-valued a_partition of n}
= card {p where p is Element of NAT*:p is one-to-one a_partition of n}

(29)

However, if we register in the Mizar environment, that there exists a set of all
a_partition of n (for more details see [7]), we can represent (29) in more
elegant form, used in [15, (16)], presented below:

card the set of all p where p is odd-valued a_partition of n
= card the set of all p where p is one-to-one a_partition of n

(30)

We can compare the obtained formulation of Euler's partitions theorem in the
Mizar system with the formulation used in the HOL system:

let EULER_PARTITION_THEOREM = prove

(`FINITE {p | p partitions n /\ !i. p(i) <= 1} /\

FINITE {p | p partitions n /\ !i. ~(p(i) = 0) ==> ODD i} /\

CARD {p | p partitions n /\ !i. p(i) <= 1} =

CARD {p | p partitions n /\ !i. ~(p(i) = 0) ==> ODD i}`

5 Conclusion

In this paper we presented a formalization of a textbook theorem where we not
only proved this theorem, but primarily tried to re�ect the main idea of the
informal proof with expressions that are available in a formal environment. We
have created more expressive de�nitions and we extended existing ones to mirror
informal mathematical language constructions in formal terms, working around
the Mizar system's limitations, without modi�cation of the core Mizar system.
We have showed that an accurate formal re�ection of informal terms obliges not
only to introduce new concepts in our library, but also to conduct reasoning in
a more di�cult way. Our studies highlighted the di�erences between the way of

conducting human legible reasoning and reasoning that is acceptable for a proof
checker.

We have showed that appropriate use of �exary operators in formal reasonings
can increase the legibility of obtained proof scripts, in e�ectively the same way
as in mathematical textbook. However, based on the same example we have
also showed a negative consequence of this method, namely the growth of the
reasoning length. Obviously, going beyond the point of this study and modifying
the core of the Mizar system, we can obtain a more natural de�nition of �exary
operators, where we do not have to explicitly use the summed sequence.

Our e�ort allowed us to formulate similarly formal equivalents of the great
majority of informal terms. Still a number of corner cases resisted our e�orts.
In particular handling sequences of length zero was problematic and we had
to fall back to non-uniform treatment of the zero-length case for the sequence
2α1 + 2α2 + . . .

We believe that this study brings us closer to the situation that informal
reasonings can be conducted in systems such as Mizar.

